



xxter script examples

Example 1: Delay a command

Example 2: Fading RGB lights

Example 3: Wake up light

Example 4: Presence simulation

Example 5: Daylight synchronized spot



Example 1: Delay a command

```
#Example for a delay of a command
WAIT 10 MIN
SET BYTE(21/Office delayed) TO BYTE(22/Office)
#Add a trigger that starts this script with the appropriate action (Office)

#To add a blocking function for this script, create another script that enables
or disables this script.
#Add a trigger for this script with the appropriate action (Block Office delay)
SET SCRIPT(3/Delay example) TO BIT(4/Block Office delay)
```

Example 2a: Fading RGB lights 1

```
#Example for fading RGB lights 1
#This example fades between 3 colours that are set in this script with a fixed
speed.
FADE RGB(505/Colour wall) TO 255,0,0 IN 20 SEC STEP 100 #full red
FADE RGB(505/Colour wall) TO 0,255,0 IN 20 SEC STEP 100 #full green
FADE RGB(505/Colour wall) TO 0,0,255 IN 20 SEC STEP 100 #full blue
RESTART
#You can use a trigger or a scenario to start or stop this script
#For example, start this script with a scenario and stop it with the Room Off
button (trigger)
```

Example 2b: Fading RGB lights 2

```
#Example for fading RGB lights 2
#This example fades between 3 colours that can be set with an iOS or Android
device and the speed is a dimmer value which can also be altered on the device.
FADE RGB(505/Colour wall) TO RGB(501/RGB colour1) IN BYTE(500,RGB fade speed)
SEC STEP 100
FADE RGB(505/Colour wall) TO RGB(502/RGB colour2) IN BYTE(500,RGB fade speed)
SEC STEP 100
FADE RGB(505/Colour wall) TO RGB(503/RGB colour3) IN BYTE(500,RGB fade speed)
SEC STEP 100
RESTART
#You can use a trigger or a scenario to start or stop this script
#For example, start this script with a scenario and stop it with the Room Off
button (trigger)
```



Example 3: Wake up light

```
#Example for creating a 'wake up' light
#This example fades a light upwards slowly
FADE BYTE(400/Bedroom wall) TO 5 IN 1 MIN
FADE BYTE(400/Bedroom wall) TO 20 IN 2 MIN
FADE BYTE(400/Bedroom wall) TO 50 IN 3 MIN
FADE BYTE(400/Bedroom wall) TO 100 IN 4 MIN
#You can use a schedule to start this script, it will stop after it has
finished
```

Example 4: Presence simulation

```
#Example for presence simulation
#This example consists of 4 scripts.

#Script 1 for starting the simulation
#This script turns two schedulers ON which start scripts 3 and 4 (in the
morning and evening).
#But only if there has not been anyone at home for more than 6 hours...
WAIT 6 HOUR
SET SCHEDULER(21/Presence morning) TO 1
SET SCHEDULER(22/Presence evening) TO 1
#You could use a trigger on the 'all off' near the front door to start this
script or a trigger on a contact of the burglar alarm, so the script is started
when the alarm is turn on.

#Script 2 for stopping the simulation
#This script turns the schedulers off and stops the starting script.
SET SCHEDULER(21/Presence morning) TO 0
SET SCHEDULER(22/Presence evening) TO 0
STOP SCRIPT(1/Start presence simulation)
STOP SCRIPT(3/Morning simulation)
STOP SCRIPT(4/Evening simulation)
#You could use a presence detector to execute this script, or trigger it when
the burglar alarm is turned off.

#Script 3 the simulation itself for the morning
WAIT 0 RANDOM 30 MIN
#First we wait for a random time, so every day the time the simulation start is
a little different.
SET BYTE(25/Bedroom wall light) TO 50
SET BIT(23/Hallway) TO 1
WAIT 5 MIN
SET BYTE(11/Kitchen light) TO 80
WAIT 15 MIN RANDOM 10 MIN
SET BYTE(25/Bedroom wall light) TO 0
SET BYTE(14/Living room spots) TO 80
SET BIT(80/Curtains) TO 0
WAIT 20 MIN RANDOM 20 MIN
SET BIT(23/Hallway) TO 0
SET BYTE(11/Kitchen light) TO 0
SET BYTE(14/Living room spots) TO 0
```



```
#Script 4 the simulation for the evening
WAIT 0 RANDOM 1 HOUR
#First we wait for a random time, so every day the time the simulation start is
a little different.
SET BYTE(11/Kitchen light) TO 70
SET BIT(23/Hallway) TO 1
SET BYTE(14/Living room spots) TO 80
WAIT 30 MIN RANDOM 30 MIN
SET BYTE(11/Kitchen light) TO 0
SET BYTE(14/Living room spots) TO 70
SET BYTE(24/Living room standing) TO 40
WAIT 1 HOUR RANDOM 1 HOUR
SET BIT(80/Curtains) TO 1
SET BYTE(25/Bedroom wall light) TO 70
SET BYTE(14/Living room spots) TO 50
WAIT 1 HOUR RANDOM 1 HOUR
SET BIT(23/Hallway) TO 0
SET BYTE(24/Living room standing) TO 0
SET BYTE(14/Living room spots) TO 0
WAIT 5 MIN
SET BYTE(25/Bedroom wall light) TO 0
```

Example 5: Daylight synchronized spot

```
#Example for synchronizing spots with the daylight
#This script starts by calculating the maximum sunlight intensity from 3 sides:
east, south and west.
$MAXSUN = 2BYTE(3/Sunlight East)
IF 2BYTE(4/Sunlight South)>$MAXSUN
  $MAXSUN = 2BYTE(4/Sunlight South)
IF 2BYTE(5/Sunlight West)>$MAXSUN
  $MAXSUN = 2BYTE(5/Sunlight West)
#With this maximum we use a formula to reduce this number to a reasonable
percentage. For this example we calculate this using the square root of the
maximum, divided by 4.
$LIGHTX = ($MAXSUN ^ (1/2))/4
#Maximize the level to 100
IF ($LIGHTX>100)
  $LIGHTX = 100
#Set the light to the calculated level
SET BYTE(21/Spots) TO $LIGHTX
#Wait for 1 minute and then restart
#This synchronises the spot's light level to the sun intensity
WAIT 1 MIN
RESTART
```